TEPT5700B

QSPICE Simulation Model — Help File

Vishay TEPT5700 Ambient Light Sensor • Rev 1.7 • Doc No. 81321

Hybrid Behavioural SPICE Model

Files

Put all files in the same folder. QSPICE finds them automatically — no path configuration needed.

File	Purpose
TEPT5700B.cir	The model subcircuit. Never run this file directly — it contains only a .SUBCKT definition with no simulation command. Running it directly causes: Fatal Error: No analysis request found.
simOPb.cir	Operating point check. Verifies the model loads correctly. Run this first.
simlux.cir	Illuminance sweep. Sweeps EV pin 1→1000 lux. Reproduces datasheet Fig. 5.
simWL.cir	Spectral sweep. Sweeps WL pin 350→1100 nm. Reproduces datasheet Fig. 7.
simT.cir	Temperature sweep. Sweeps TK pin 233→373 K (−40→+100 °C). Reproduces datasheet Fig. 3.

⚠ TEPT5700B.cir is a library file, not a simulation. Always open and run one of the sim*.cir files. They include TEPT5700B.cir automatically.

Pin map

The model subcircuit has six pins. The instance line in every sim file is:

```
* X<name>  C     B       E   EV    WL    TK    <model>
X1         Cn    FloatB  0   EVn   WLn   TKn   TEPT5700
```

Pin	Function
`C`	Collector — connect to load resistor. Current flows from VCC through RL into this node.
`B`	Base — leave floating in photo mode. Connect R_float (100 MΩ) to GND to prevent singular matrix error.
`E`	Emitter — connect to GND (node 0).
`EV`	Illuminance input. 1 V = 1 lux. Drive with a DC voltage source or sweep source.
`WL`	Wavelength input. 1 V = 1 nm. Example: 570 V = 570 nm (peak sensitivity).
`TK`	Temperature input. 1 V = 1 Kelvin. Example: 298.15 V = 25 °C.

Step 1 — Operating point check (simOPb.cir)

Run this first every time you open the project. It confirms the model file is found and the DC bias is correct.

The netlist

```
* TEPT5700 operating point check
* Expected result: I(RL) = 75 uA

.INCLUDE TEPT5700B.cir

* X<name>  C     B       E   EV    WL    TK    <model>
X1         Cn    FloatB  0   EVn   WLn   TKn   TEPT5700

VCC  VCC  0   DC 5       ; 5V supply
RL   VCC  Cn  1k          ; 1 kΩ load

VEV  EVn  0   DC 100     ; 100 lux
VWL  WLn  0   DC 570     ; 570 nm peak wavelength
VTK  TKn  0   DC 298.15  ; 25 deg C in Kelvin

R_float  FloatB  0  100MEG  ; prevents floating node error

.MEAS OP IPCE FIND I(RL)
.OP
.END
```

How to run

1. Open simOPb.cir in QSPICE (File → Open).
2. Press F5.
   The simulation takes under 0.1 seconds.
3. Read the Output Window at the bottom of the QSPICE screen.

Expected Output Window result

```
IPCE = -7.50082e-05
```

 The negative sign is normal — QSPICE measures I(RL) as current from VCC toward Cn. The photocurrent flows the other way. The magnitude 75 µA is what matters.

What to check

Output / Error	Meaning and action
`magnitude ≈ 75 µA`	Model loaded correctly. Datasheet typ: 75 µA at 100 lux, 570 nm, 25 °C.

Output / Error	Meaning and action
`"No analysis request"`	You ran TEPT5700B.cir directly. Open simOPb.cir instead.
`"Subckt not found"`	TEPT5700B.cir is not in the same folder as simOPb.cir.
`"Floating node"`	R_float line is missing or deleted. Check it is present.
`I(RL) ≈ 5 nA`	EV source is 0 V (dark). Check VEV line reads DC 100.

Step 2 — Illuminance sweep (simlux.cir)

Sweeps the EV pin from 1 to 1000 lux at fixed wavelength (570 nm) and temperature (25 °C). Reproduces datasheet Fig. 5.

The netlist

```
* TEPT5700 illuminance sweep

.INCLUDE TEPT5700B.cir

X1          Cn   FloatB  0    EVn    WLn    TKn    TEPT5700

VCC  VCC  0    DC 5
RL   VCC  Cn   1k

VEV  EVn  0    DC 100      ; swept: 1 to 1000 lux
VWL  WLn  0    DC 570
VTK  TKn  0    DC 298.15

R_float  FloatB  0  100MEG

.TEMP 25
.DC VEV 1 1000 1
.meas lux20   find I(rl) at 20
.meas lux100  find I(rl) at 100
.meas lux1000 find I(rl) at 1000
.END
```

⚠ simlux.cir must use .INCLUDE TEPT5700B.cir — note the B suffix. If it reads TEPT5700.cir (no B) the simulation will fail with subckt not found.

Expected Output Window results

```
lux20   = -1.19e-05    (11.9 µA — inside datasheet band 5.2-24 µA)
lux100  = -7.50e-05    (75.0 µA — datasheet typ)
lux1000 = -7.55e-04    (755 µA  — linear extrapolation)
```

Viewing the waveform — log-log plot matching datasheet Fig. 5

4. Press F5 to run.

5. Click anywhere on the simOPb.cir text window to bring it into focus.
6. The waveform viewer opens. Right-click inside it → Add Trace → select I(RL). Click OK.
7. Right-click the X axis → Set Axis Range → tick Log scale.
8. Right-click the Y axis → Set Axis Range → tick Log scale.

Illuminance reference values

EV (V = lux)	Typical environment
`1 lux`	Moonlit night
`20 lux`	Very dim room. Datasheet band: 5.2–24 µA
`100 lux`	Well-lit office. Datasheet typ: 75 µA
`500 lux`	Bright indoor / near window
`1000 lux`	Overcast outdoor daylight

Step 3 — Spectral sweep (simWL.cir)

Sweeps the WL pin from 350 to 1100 nm at fixed illuminance (100 lux) and temperature (25 °C). Reproduces datasheet Fig. 7.

The netlist

```
* TEPT5700 spectral sweep

.INCLUDE TEPT5700B.cir

X1          Cn   FloatB   0    EVn    WLn    TKn    TEPT5700

VCC   VCC   0    DC 5
RL    VCC   Cn   10k         ; 10 kΩ gives more voltage swing for low-light
wavelengths

VEV   EVn   0    DC 100
VWL   WLn   0    DC 570      ; swept: 350 to 1100 nm
VTK   TKn   0    DC 298.15

R_float   FloatB   0   100MEG

.DC VWL 350 1100 5
.meas WL440nm find I(rl) at 440
.meas WL570nm find I(rl) at 570
.meas WL800nm find I(rl) at 800

* Note: no .END line — QSPICE accepts this but adding .END is good
practice
```

Expected Output Window results

```
WL440nm = -1.34e-05     (13.4 µA  — relative sensitivity 0.178)
WL570nm = -7.52e-05     (75.2 µA  — peak, relative = 1.000)
WL800nm = -4.09e-06     ( 4.1 µA  — relative sensitivity 0.055)
```

Displaying relative sensitivity in the waveform viewer

To plot S_rel (0 to 1) matching datasheet Fig. 7, type this expression into the Add Trace dialog instead of selecting I(RL) from the list:

```
I(RL) / 7.52e-5

* Divides every point by the peak current at 570 nm.
* The curve then reads 1.0 at 570 nm and falls off symmetrically.
```

Spectral reference values

WL (V = nm)	S_rel model	Colour / region
350 nm	0.00	UV — no response
440 nm	0.18	Violet — λ0.5 lower edge
490 nm	0.50	Blue-green — half power
570 nm	1.00	Yellow-green — peak (λp)
660 nm	0.44	Red
750 nm	0.09	Near-IR
800 nm	0.05	Near-IR — λ0.5 upper edge
1000 nm	0.00	IR — no response

Step 4 — Temperature sweep (simT.cir)

Sweeps the TK pin from 233.15 K (−40 °C) to 373.15 K (+100 °C) at fixed illuminance (100 lux) and wavelength (570 nm). Reproduces datasheet Fig. 3.

The netlist

```
* TEPT5700 temperature sweep

.INCLUDE TEPT5700B.cir

X1   Cn  FloatB  0   EVn  WLn  TKn  TEPT5700

VCC  VCC  0   DC 5
RL   VCC  Cn  1k

VEV  EVn  0   DC 100
VWL  WLn  0   DC 570
VTK  TKn  0   DC 298.15  ; swept: 233.15 to 373.15 K

R_float  FloatB  0  100MEG

.DC VTK 233.15 373.15 2

.meas Tm40C  find -I(RL) at 233.15   ; -40 deg C
.meas T25C   find -I(RL) at 298.15   ;  25 deg C
.meas T85C   find -I(RL) at 358.15   ;  85 deg C
.END
```

simT.cir uses -I(RL) in the .meas lines (negative sign) so results appear as positive numbers in the Output Window. simWL.cir and simlux.cir use I(RL) without the minus, so those results appear negative — both are correct.

Expected Output Window results

```
Tm40C = 5.25e-05    (52.5 µA  — 0.700x relative to 25 °C)
T25C  = 7.50e-05    (75.0 µA  — 1.000x reference)
T85C  = 1.17e-04    (117 µA   — 1.550x relative to 25 °C)
```

Displaying relative photocurrent matching datasheet Fig. 3

In the Add Trace dialog, type:

```
-I(RL) / 7.50e-5

* Divides by the 25 deg C reference value.
* X axis = V(TKn) in Kelvin. To show Celsius: use the expression V(TKn)-
273.15
* as a secondary axis label (QSPICE does not support axis relabelling
directly).
```

Temperature reference values

TK (V = Kelvin)	Relative IPCE
233.15 K (−40 °C)	0.700× — minimum operating temperature (datasheet Abs Max)
273.15 K (0 °C)	0.848×
298.15 K (25 °C)	1.000× — all datasheet specs reference this temperature
333.15 K (60 °C)	1.221×
358.15 K (85 °C)	1.550× — maximum operating temperature (datasheet Abs Max)
373.15 K (100 °C)	1.830× — maximum storage temperature

Model internals (TEPT5700B.cir)

The model uses eight elements — no NPN transistor. All photocurrent is generated directly by behavioral sources (B-sources). This avoids transistor beta nonlinearity which caused up to 80% current error at off-peak wavelengths in earlier model versions.

Element	Function
`B_dTc`	Temperature deviation from 25 °C. Feeds B_KTC and B_IDARK. dTc = V(TK) − 298.15
`B_ARG1`	Gaussian exponent for main spectral lobe: (WL−570)²/9800
`B_ARG2`	Gaussian exponent for near-IR tail: (WL−800)²/10000
`B_SPEC`	Relative spectral sensitivity: min(1, exp(−arg1) + 0.050×exp(−arg2))
`B_KTC`	Temperature multiplier from datasheet Fig. 3. Equals 0.700 at −40 °C, 1.000 at 25 °C, 1.550 at 85 °C.
`B_IPH`	Main photocurrent: 7.501e-7 × EV × S(WL) × Ktc(TK). Calibrated to 75 µA at 100 lux, 570 nm, 25 °C.
`B_IDARK`	Dark current: 5 nA at 25 °C, doubles every 10 °C. Uses exp() not pow().
`D1 DSAT`	Saturation diode. Clamps VCE above ~0.1 V, modelling VCEsat = 0.1 V (datasheet page 2).
`C_CCEO`	Collector-emitter capacitance 16 pF (datasheet Fig. 6 at VCE = 0).
`R_BASE`	100 MΩ base termination. Leave B pin unconnected in normal use.

Why B-sources are split into separate nodes

Early model versions put the entire spectral expression on one B_SPEC line. QSPICE evaluated the nested `exp()` calls inconsistently at large arguments, giving wrong values at 800 nm. The fix was to split the exponent calculation into separate intermediate nodes (`arg1`, `arg2`) so each B-source evaluates a simple expression. The math is identical but the evaluation is unambiguous.

Sign convention — why I(RL) is negative

In SPICE, `I(RL)` measures current flowing from the first node to the second node *through* the element. RL is defined as:

```
RL   VCC  Cn  1k    ; first node = VCC, second node = Cn
```

So `I(RL)` measures current flowing from VCC toward Cn. The photocurrent actually flows from Cn upward through RL to VCC — the opposite direction — so QSPICE reports it as negative.

To get positive numbers, either:

- Use
  -I(RL)

- Swap the node order:
  RL Cn VCC 1k

 simT.cir uses -I(RL) throughout so temperature results are positive. simWL.cir and simlux.cir use I(RL) so results are negative. The magnitudes are always correct.

Troubleshooting

"Fatal Error: No analysis request found"

You opened and ran `TEPT5700B.cir` directly. It is a subcircuit definition — it has no `.OP`, `.DC`, or `.END` command. Open one of the `sim*.cir` files instead.

"Subckt TEPT5700 not found"

- TEPT5700B.cir is not in the same folder as the sim file.
- The .INCLUDE line in simlux.cir reads TEPT5700.cir (no B suffix). Change it to TEPT5700B.cir.
- The filename on disk is wrong — check Windows Explorer with file extensions visible (View → File name extensions).

"Singular matrix" / "Floating node"

The `R_float  FloatB  0  100MEG` line is missing from the sim file. Every sim file must include this line. Without it, the B (base) pin of the model has no DC path to ground and QSPICE cannot solve the matrix.

Output Window shows nothing after .OP

The `.OP` directive alone does not print to the Output Window. You need a `.MEAS OP` line to extract a value. simOPb.cir already has:

```
.MEAS OP IPCE FIND I(RL)
```

If you deleted this line, add it back just before `.OP`.

Waveform viewer is blank after sweep

Click on the *sim file text window* (not the log window) before opening the waveform viewer. The viewer only activates when a simulated netlist window has focus. Then right-click inside the plot area → Add Trace.

.DC sweep shows only one point

Check the `.DC` line format exactly:

```
.DC  <source_name>  <start>  <stop>  <step>

* Examples from the sim files:
.DC VEV 1 1000 1       ; sweeps source named VEV from 1 to 1000 in steps
of 1
.DC VWL 350 1100 5     ; sweeps source named VWL from 350 to 1100 in steps
of 5
```

```
.DC VTK 233.15 373.15 2
```

The source name in the `.DC` line must exactly match the voltage source name in the netlist (`VEV`, not `EVn`).

.meas AT= gives wrong value

The `AT=` syntax only works with `.DC` sweeps, not with `.OP`. Use `FIND I(RL) AT 100` (space, not equals) in `.DC` contexts. Use `FIND I(RL)` without AT= for `.OP`.

Quick reference

TK pin — temperature conversion

VTK (Volts = Kelvin)	Temperature
233.15	−40 °C (minimum operating temperature)
253.15	−20 °C
273.15	0 °C
298.15	25 °C ← all datasheet specs use this value
333.15	60 °C
358.15	85 °C (maximum operating temperature)
373.15	100 °C (maximum storage temperature)

EV pin — common illuminance levels

VEV (Volts = lux)	Environment
0	Complete darkness — tests dark current only
1	Moonlit night
20	Datasheet test point — IPCE band 5.2–24 µA
100	Datasheet reference — IPCE 75 µA typ
1000	Overcast outdoor daylight
10000	Direct sunlight (use RL ≤ 47 Ω)

WL pin — key wavelengths

VWL (Volts = nm)	Colour / response
440	Violet — lower spectral edge S ≈ 0.18
490	Blue-green — half power point S ≈ 0.50
570	Yellow-green — peak sensitivity S = 1.00
620	Orange

VWL (Volts = nm)	Colour / response
660	Red S ≈ 0.44
800	Near-IR — upper spectral edge S ≈ 0.05

Load resistor guidelines

RL value	Use case
1 kΩ	Default for low-to-medium illuminance (< 500 lux). VCE stays > 1V.
10 kΩ	Used in simWL.cir for spectral sweep — gives more voltage swing at low current wavelengths.
47 Ω	Required for direct sunlight (> 5000 lux) to keep VCE in active region.

Model parameters from Vishay TEPT5700 Datasheet Rev 1.7, Doc No. 81321. For QSPICE (Qorvo).